

TUTORIAL

NONMEM Tutorial Part I: Description of Commands and Options, With Simple Examples of Population Analysis

Robert J. Bauer^{1,*}

In this tutorial, the various components of NONMEM will be described, and the basic steps of setting up NONMEM control stream files and data files will be demonstrated. Some basic concepts of nonlinear mixed effects modeling will be discussed, along with simple examples demonstrating how to use NONMEM to perform population analysis of clinical data.

NONMEM stands for NONlinear Mixed Effects Modeling. NONMEM is a computer program that is implemented in Fortran90/95.¹ It solves pharmaceutical statistical problems in which within-subject and between-subjects variability is taken into account when fitting a pharmacokinetic and/or pharmacodynamic (PK/PD) model to data. The development and application of PK/PD models by pharmaceutical scientists using the appropriate statistical methodology can provide a basis for determining optimal dosing strategies for their products and increases their understanding of drug mechanisms and interactions.

The NONMEM software was originally developed by Lewis Sheiner and Stuart Beal and the NONMEM Project Group at the University of California and has been used for more than 30 years for population analysis by many pharmaceutical companies and the PK/PD modeling community. Its continued development and improvement assure that they may continue to use the analysis tool with which they are familiar for present-day pharmaceutical development.

The PK/PD data are typically collected from clinical studies of pharmaceuticals agents involving the administration of a drug to individuals and the subsequent observation of drug, metabolite, and/or biomarker levels (most often in the blood, plasma, or urine) as well as clinical outcome measures. The proper modeling of these data involves accounting for both unexplainable between-subject (interindividual) and within-subject (intraindividual) variability via random effects as well as PK and PD parameters and subject factors via fixed effects terms. This mixed effects modeling is especially useful when there are only a few measurements from each individual sampled in the population or when the data collection design varies considerably between these individuals. NONMEM is a general program that can be used to fit models to a wide variety of data.

BASIC STRUCTURE AND EXECUTION OF NONMEM

The components of the NONMEM software consist of the following (diagrammed in **Figure 1**):

- **NM-TRAN (NonMem TRANslator)**—converts the data file and the control stream into FORTRAN code files for use by NONMEM (guide IV¹)
- **PRED (PREdiction)**—computes predictions of observations and partial derivatives
- **PREDPP (PREd for Population Pharmacokinetics)**—computes the predictions of observations arising from one of several specialized PK models in a library of PK model subroutines. There are also some general models. Several models (**ADVANS**) and reparameterizations of these models (**TRANS**) are available (see **Table 1** and guide VI¹).
- **NONMEM (NONlinear Mixed Effects Modeling)**—uses various statistical methodologies to provide estimates of parameters (and their statistical uncertainties) that result in best fit of the model predictions to the observed data.

NONMEM may be executed using a command line or via third-party software tools, such as PDx-Pop,² Pirana,³ Perl Speaks NONMEM⁴, KIWI,⁵ PLT Tools,⁶ or Wings for NONMEM,⁷ that execute the command line for you. The NONMEM guides¹ provide theoretical descriptions and reference information on how to use the software, but do not offer tutorial-style development. This tutorial will offer a brief but direct outline of how to use NONMEM with population analysis examples, explaining all the instruction records and data file setups. A full textbook of working with NONMEM and the classical methods with examples is provided in ref.⁸ The advanced tutorial will describe the following various estimation methods: first order, first-order conditional, Laplace, expectation-maximization, and Bayes methods.

We begin with the most basic manner of executing NONMEM with a simple command implemented from a DOS Windows or Linux (or Apple OSX) command terminal:

Windows:

```
nmfe74 filename.ctl filename.res
```


Linux:

```
./nmfe74 filename.ctl filename.res
```

The nmfe74 script sets up the NONMEM problem. The required options are:

¹Pharmacometrics, R&D, ICON Clinical Research LLC, Gaithersburg, MD, 20878. *Correspondence: Robert J. Bauer (Robert.Bauer@iconplc.com)

Received: January 16, 2019; accepted: March 10, 2019. doi:10.1002/psp4.12404



The typical NONMEM operating environment

Figure 1 Relationship between NM-TRAN, NONMEM, and PREDPP. The diagram shows the workflow executed by the `nmfe74` script. The user's control stream is submitted to NM-TRAN, which then renders the code into a Fortran file (FSUBS), transfers the information in the data file to FDATA, and transfers option settings to file FCON. NM-TRAN also sets up memory allocation settings (which can be viewed in FSIZES). The `nmfe74` script then compiles FSUBS, calls in appropriate PREDPP subroutines for linking, builds the NONMEM executable using the memory allocation specifications, and prepares a parallel computing setup if necessary. The script then executes the NONMEM executable. NONMEM, using the PREDPP built-in model routines, performs estimations and statistical calculations and produces output files containing the various results.

Control stream: user-supplied modeling & estimation instruction file for NONMEM, typically named as *filename.ctl*. An example control stream file will be shown later.

Result file: The file that records the main results of the run (also called NONMEM report file). The convention is to use the same root names as the control stream, with `.res` or `.lst` extension.

Additional options may be added to control the build or execution environment, such as to request default precompiled PREDPP model routines (`-prdefault`) for a faster build or to parallelize the run (`-parafile=...`) to distribute computation effort across two or more computer cores for faster execution.

The script first calls the NONMEM translator program NM-TRAN (guide IV of ref.¹), which converts user-inputted options, instructions, data, and code into usable form and places these into specially formatted files. NM-TRAN then builds the NONMEM executable according to these specifications (sets up array sizes, etc.), incorporating the appropriate model type, and the script then continues by executing NONMEM. The NONMEM program reads the specially formatted files prepared by NM-TRAN. The particular built-in

model type (PREDPP) is executed by the modeling system within NONMEM. These built-in models are listed in **Table 1**, called ADVANs. More information about these ADVANs can be found in NONMEM guide VI.¹

The control stream may call additional input files to be used. One of these is the data file, typically named as *filename.csv*. This is a user-supplied text file containing all of the data that is translated by NM-TRAN to a usable data set (FDATA) to be used by NONMEM.

CREATING A DATA FILE

The data file is a text file that is comma or space delimited and consists of separate dosing and observation records containing a value for each of the data items (chapter 6, guide V¹). A data record is defined on a single line of the data file, and the data item is defined in the column of the data file. NONMEM (via NM-TRAN and PREDPP) is extremely versatile when it comes to handling dosing and observations in a data set.

NONMEM data items

When one of the PREDPP ADVAN models is used, data items ID, AMT, TIME, and DV are required.

ID, subject (individual) identification number. All records for a given subject must be placed together.

TIME, the recorded times of dosing events and/or observations, such as blood sample collection to record concentrations or PD end-point measurements. These TIME values must be sorted in ascending order for a given ID until the end of the subject's records or at a record with `EVID = 3` or `EVID = 4` (see EVID below) that specifies a new occasion.

AMT, dose administered for dosing record, or zero or a period as a place holder for observation records

DV, the dependent variable or observed concentration, or zero or a period as a place holder on a dosing record

Additional columns may be added for:

RATE, rate of infusion of drug. `RATE = 0` if drug is given as a bolus.

MDV, missing DV.

`0` = not missing data value, and the DV of the record is to be used in the analysis.

`1` = missing data value. Setting MDV to 1 would be appropriate to evaluate the prediction value at a nonobservation TIME or for the DV values of this record not to be included in the analysis.

EVID, event ID:

`0` = observation. The DV value of this record will be used in the estimation if MDV = 0.

`1` = dose. The AMT, II, RATE, and ADDL of this record will be used for the drug input information. Dose records for each dose event may occur throughout the individual's data set.

`2` = other. The user may use this in his or her model to trigger a special event, such as:

IF(EVID == 2) AUC = A(5)

`3` = reset, indicates start of a new occasion (do not use when starting the first occasion). All compartment values in the model will be reset to 0. For this record, the user may also set the TIME to 0 or other convenient starting value for this occasion of the individual's data.

Table 1 List of built-in models as of NONMEM 7.4 (PREDPP guide VI¹)

Model type subroutine	Compartments	Parameterization subroutine	Basic parameters	Model description
ADVAN1	1=CENTRAL 2=OUTPUT	TRANS1	K, V	One-compartment linear model
ADVAN2	1=DEPOT 2=CENTRAL 3=OUTPUT	TRANS2	CL, V	One-compartment linear model with first-order absorption
		TRANS1	KA, K	
ADVAN3	1=CENTRAL 2=PERIPHERAL 3=OUTPUT	TRANS2	CL, V, KA	Two-compartment linear mammillary model
		TRANS1	K, K12, K21	
		TRANS3	CL, V, Q, VSS	
		TRANS4	CL, V1, Q, V2	
		TRANS5	AOB, ALPHA, BETA	
ADVAN4	1=DEPOT 2=CENTRAL 3=PERIPHERAL 4=OUTPUT	TRANS6	ALPHA, BETA, K21	Two-compartment linear mammillary model with first-order absorption
		TRANS1	KA, K, K23, K32	
		TRANS3	CL, V, Q, VSS, KA	
		TRANS4	CL, V2, Q, V3, KA	
		TRANS5	AOB, ALPHA, BETA, KA	
		TRANS6	ALPHA, BETA, K32, KA	
ADVAN10	1=CENTRAL 2=OUTPUT	TRANS1	VM, KM	One-compartment model with Michaelis-Menten elimination
ADVAN11	1=CENTRAL 2=PERIPHERAL 1 3=PERIPHERAL 2 4=OUTPUT	TRANS1	K, K12, K21, K13, K31	Three-compartment linear mammillary model
		TRANS4	CL, V1, Q2, V2, Q3, V3	
		TRANS6	ALPHA, BETA, GAMMA, K21, K31	
ADVAN12	1=DEPOT 2=CENTRAL 3=PERIPHERAL 1 4=PERIPHERAL 2 5=OUTPUT	TRANS1	KA, K, K23, K32, K24, K42	Three-compartment linear mammillary model with first-order absorption
		TRANS4	CL, V2, Q3, V3, Q4, V4, KA	
		TRANS6	ALPHA, BETA, GAMMA, KA, K32, K42	
ADVAN5	GENERAL LINEAR MODEL	TRANS1		General linear model (with real or complex eigenvalues)
ADVAN7	GENERAL LINEAR MODEL WITH REAL EIGENVALUES	TRANS1		General linear model with real eigenvalues
ADVAN6	GENERAL NONLINEAR MODEL: RK56 ODE SOLVER (NONSTIFF)	TRANS1		General nonlinear model (DVERK1)
ADVAN8	GENERAL NONLINEAR MODEL: GEAR ODE SOLVER	TRANS1		General Nonlinear Model with Stiff Differential Equations (DGEAR1)
ADVAN9	GENERAL NONLINEAR MODEL: LSODI ODE SOLVER, AND DIFFERENTIAL-ALGEBRAIC EQUATIONS (ADE)	TRANS1		General nonlinear model with equilibrium compartments (ordinary and algebraic differential equations, LSODI1)
ADVAN13	GENERAL NONLINEAR MODEL: LSODA ODE SOLVER	TRANS1		General nonlinear model with stiff or non-stiff differential equations (LSODA)
ADVAN14	GENERAL NONLINEAR MODEL: CVODES ODE SOLVER	TRANS1		General nonlinear model with stiff non-stiff differential equations (CVODES)
ADVAN15	GENERAL NONLINEAR MODEL: IDAS ODE AND ADE SOLVER	TRANS1		General nonlinear model with equilibrium compartments (ordinary and algebraic differential equations, IDAS)

K=rate constant of elimination; V=central volume; Kxy=rate constant of transfer from compartment x to compartment y; Vx=compartment x volume; CL=clearance; KA=rate constant of absorption; Q=intercompartmental clearance; VSS=volume of distribution at steady state; AOB=A over B, macroconstant; ALPHA=alpha (first) exponent, macroconstant; BETA=beta (second) exponent, macroconstant; GAMMA=gamma (third) exponent, macroconstant

4=reset & dose, indicates start of a new occasion (do not use when starting the first occasion). All compartment values in this model will be reset to 0, and a dose of the amount AMT will be added. For this record, the user may also set the TIME to 0 or other convenient starting value for this occasion of the individual's data.

With PREDPP, the data preprocessor (NM-TRAN) will generate default values for EVID and MDV.

CMT, compartment number (see Table 1), specified for a dose or observation, not required if defaults are used.

SS, steady-state dosing.

0=non-steady-state dosing.

1=steady-state dosing (its use is shown later).

ADDL, additional identical doses given, which is 1 less than total number of identical doses.

II, interdose interval, time between additional doses.

Data items pertaining to covariate factors may be included in the data set (e.g., weight, height, age, body surface area, serum creatinine, etc.)

General rules for creating data files

Data files can be space-delimited or comma-delimited ASCII text. You may also use tab, but some third-party softwares that interact with NONMEM may not allow it.

The name may be any valid filename with or without path, but it is safest to have file names and paths without spaces.

List all records from one individual sequentially. Do not intersperse individuals.

Every record must have at least a placeholder for each item. Use 0 or "." for null for placeholders, as the "." is treated as 0 during translation and execution.

Use a separate record for each event (dose, observation, etc.).

All records must be ordered chronologically by actual event time within an individual and occasion. Dose records may be added with EVID = 1 throughout an occasion, their effect added to the compartment contents from the effect of previous doses. Upon the beginning of a new occasion after the first, with EVID = 3 or EVID = 4, compartment values are reset to 0, and TIME values are permitted to be reset to a new starting value (usually 0) as well.

Coding dosing events. Dosing is coded in the data set via dosing records consisting of combinations of the TIME, AMT, RATE, CMT, ADDL, and II data items appearing in an individual's data in the order in which they occurred in the clinical setting and are translated appropriately by NM-TRAN. (The default dosing CMT is assumed to be 1 and ordinarily need not always be coded.) The summary of various dose events is as follows:

DOSAGE FORM	TIME	AMT	RATE	CMT(=1)	ADDL (IF MULTIPLE)	II (IF MULTIPLE)
IV BOLUS	X	> 0	0	CENTRAL	> 0	> 0
IV INFUSION	X	> 0	> 0	CENTRAL	> 0	> 0
ORAL	X	> 0	0	DEPOT	> 0	> 0
IM	X	> 0	0	DEPOT	> 0	> 0

Doses (AMT) and concentrations (DV) do not have to be in equivalent mass units (e.g., 100 mg dose, 10 µg/L concentration) To account for this, adjust the scale of the observation compartment in the control stream (e.g., S1 = V/1,000).

Example a:

```
C Data Desc: example data file
C ID AMT TIME DV CMT RATE SMK WT
. 1 80 0 0 1 10 1 72
. 1 0 2 0.5 1 0 1 72
```

This example codes an infusion of 10 mg per hour for 8 hours (80 mg over 8 hours) and an observation taken at 2 hours.

Example b:

```
C Data Desc: example data file
C ID AMT DATE TIME ADDL II DV HT
. 1 50 1 12:00 20 8 0 65
. 1 0 2 8:00 0 0 1.7 65
. 1 0 3 13:30 0 0 2.5 65
```

This example codes a 50-mg oral dose given every 8 hours for 7 days and observations taken at 8:00 AM and 1:30 PM on days 2 and 3. NM-TRAN internally converts the **actual times** given by the DATE and TIME items in this example to **elapsed times**. The time for the first line would become 0.0 and the times for lines 2 and 3 become 20.0 and 49.5 hours, respectively. For the above to represent oral dosing, the model chosen would have to specify a depot compartment, such as ADVAN2, ADVAN4, or ADVAN12. For such models, the default compartment is 1 for the dose and 2 for the observation. Or, the CMT data item may be used to explicitly state which compartment pertains to the observation or dose record.

Steady-state dosing. A steady-state dose is one that is imagined to be the last of a series of implied doses each of which is exactly like the dose being coded given at regular intervals specified by the II data item and achieving steady state by the time the steady-state dose is given. Steady-state dosage forms are summarized as follows:

Steady-state dose	AMT	RATE	SS	II
Multiple bolus or extravascular	> 0	=0	1	> 0
Multiple infusions	> 0	> 0	1	> 0
Steady-state constant infusion	=0	> 0	1	=0

(SS = 0 for non-steady-state doses or for observations.)

Some examples are:

Example c:

```
C ID TIME AMT RATE SS II
. 9 48 320 . 1 8
```

320 mg bolus every 8 hours, steady state at 48 hours.

Example d:

```
C ID TIME AMT RATE SS II
. 9 48 320 160 1 8
```

320 mg infusion given over 2 hours every 8 hours, steady state at 48 hours.

Example e:

```
C ID TIME AMT RATE SS II
. 9 48 . 20 1 .
```

20 mg/hour steady-state constant infusion, steady state at 48 hours.

A SIMPLE POPULATION ANALYSIS WITH NO COVARIATES

We are given the concentration-time data for a group of 30 patients. The patients received 250 mg of a drug administered as an intravenous bolus. The PK observations for each patient were recorded at 5, 10, 15, 30, 45 minutes, 1, 1.5, 2, 4, 6, 8, 12 hours. A two-compartment model will be used to fit the data. Models for between-subject and random residual errors will be used.

The layout of the data file 402.csv is as follows (full data set is in **Supplementary Material S1**):

First three rows of patient 1 data (a "." is treated as a 0 by NONMEM):

```
C Data Desc: Chapter 5
C ID TIME DV AMT
. 1 0 . 250
. 1 0.08 27.947 .
. 1 0.17 21.014 .
```

Last three rows of patient 1 data, and First three rows of patient 2 data:

```
. 1 6 2.6298 .
. 1 8 2.3164 .
. 1 12 1.6509 .
. 2 0 . 250
. 2 0.08 24.581 .
. 2 0.17 17.312 .
```

Last three rows of patient 2 data:

```

.      2      6      2.1484      .
.      2      8      1.9529      .
.      2     12      1.2715      .

```

ID must be contiguous but not necessarily consecutive. An observation (DV) cannot be on the same line as the dose (AMT).

File types. *.ctl* files are called *control streams*. They are used to give instructions to the NONMEM software during a run.

.res files into which NONMEM writes the results of a run. *.tab* files are table files generated by NONMEM.

NOTE: These extensions are what will be used throughout this tutorial. These extensions are not required by NONMEM.

Organizational hint. Use the same “root” name for *.ctl*, *.res*, and *.tab* files as a way to associate them together.

e.g.

control stream	100.ctl
results file	100.res
table file	100.tab

```

$PROB RUN# 402 POPULATION DATA
$INPUT C ID TIME DV AMT
$DATA 402.csv IGNORE=C
$SUBROUTINES ADVAN3 TRANS4

```

```
$PK
```

```

TVV1=THETA (1)
V1=TVV1*EXP (ETA (1))

```

```

TVCL=THETA (2)
CL=TVCL*EXP (ETA (2))

```

```

TVV2=THETA (3)
V2=TVV2*EXP (ETA (3))

```

```

TVQ=THETA (4)
Q=TVQ*EXP (ETA (4))

```

```
S1=V1
```

```

$ERROR
Y=F*(1+EPS (1))

```

```
$THETA
```

```

(0, 9.8) ; [V1]
(0, 3.7) ; [CL]
(0, 8.6) ; [V2]
(0, 31) ; [Q]

```

```
$OMEGA
```

```

0.02 ; [P]
0.02 ; [P]
0.02 ; [P]
0.02 ; [P]

```

```
$SIGMA
```

```
0.02 ; [P]
```

```

$ESTIMATION METHOD=1 MAXEVAL=9999 INTER PRINT=5 FORMAT=,1PE13.6
$COV
$TABLE ID TIME DV IPRE=CIPRED AMT CL V1 Q V2 ETA1 ETA2 ETA3 ETA4
        CWRES IRES=CIRES IWRE=CIWRES NPD NPDE ESAMPLE=1000
        NOPRINT FILE=402.tab FORMAT=s1PE12.5

```

DEFINING A MODEL THROUGH THE CONTROL STREAM

The control stream includes a series of commands that are translated by NM-TRAN for use by NONMEM. The control stream is composed of several Control Records, each initiated with “\$”. The control stream is an ASCII text file and can have any valid name.

“;” (semicolon) signifies a comment for the rest of the line and will be ignored by NM-TRAN. The following examples show how this is done. The NONMEM control stream for this example (402.ctl) is shown here:

Description of the NM-TRAN control records

\$PROB—title of the problem (see chapter 2, guide V¹).

\$INPUT—list of data items in the data set to be input into NONMEM (chapter 6, guide V¹).

Reserved item labels (items that have a special meaning to NONMEM):

C, ID, DV, MDV, TIME, EVID, AMT, RATE, SS, II, ADDL, CMT, PCMT, CALL, CONT, DATE, DAT1, DAT2, DAT3, L1, L2

Other labels may serve as synonyms to reserved name items; labels can contain letters and numerals (a letter must be first), 20-character limit.

```
$INPUT ID DOSE=AMT TIME CONC=DV WT
```

Dropping data items:

```
$INPUT ID AMT TIME DV SCR CMT RATE EVID
```

```
$INPUT ID AMT TIME DV SCR=DROP CMT RATE EVID
```

With $X=DROP$, NM-TRAN ignores item X when constructing the NONMEM data set (FDATA).

\$DATA—name of the data file to be used (can include a complete path).

The **\$DATA** record:

```
$DATA file.csv
```

or

```
$DATA file.csv IGNORE=C RECORDS=200
```

IGNORE=x allows you to comment out unwanted lines from your data set by prefacing them with x (e.g. C or #). **Ignore=@** will exclude all records that begin with a letter. For example, you can have the first column serve as a placeholder “C” on the **\$INPUT** record:

```
$INPUT C ID AMT TIME DV SCR CMT RATE EVID
```

Then, those records in the data file beginning with 0 or “.” in the first column will be read by NM-TRAN, and those records beginning with C will be ignored:

```
C My data set
C ID AMT TIME DV RATE EVID
. 1 100 0.0 0.0 20.0 1
. 1 0 5.0 6.5 0.0 0
```

RECORDS=n allows you to specify first n records to use if a truncated data set is desired.

\$SUBROUTINE—specifies which model and parameterization is to be used to fit to the data (ADVAN (model), TRANS (parameterization)); see **Table 1**, and chapter 7, guide V¹).

\$PK block (chapter 7, guide¹).

This block assigns THETAs to fixed effects parameters:

```
TVV1=THETA(1)
```

```
TVCL=THETA(2)
```

TV represent the typical value of the parameter for the population (typical value clearance TVCL, typical value volume TVV, etc.).

Also, you can specify a random effect model to structural parameters using ETA to model intersubject variability. The parameter (clearance CL, volume V, etc.), represents the parameter for an individual subject. Deviations (ETA) of independent parameters from the typical mean are assumed to be normally distributed with a mean of 0 and a variance of OMEGA. These deviations of the individual’s parameters from the typical mean can be introduced in many ways, some of which are the following:

Additive: $V = TVV + ETA(4)$

Proportional: $V = TVV * (1 + ETA(4))$

Exponential: $V = TVV * EXP(ETA(4))$

Exponential is most preferred for between-subject variability, as this ensures the individual parameter is always positive, regardless of the value of ETA, which could vary from $-\infty$ to $+\infty$.

\$ERROR block (chapter 8, guide V¹)—specifies a model for the residual within-subject error. This is the error not accounted for by the between-subject variability. Given the prediction value F that is evaluated by the ADVAN model, you can give a statistical description relating to the data (Y). Deviations EPS (epsilon) of data Y from the prediction value F are assumed to be normally distributed with a mean of 0 and a variance of SIGMA. These deviations from the prediction F can be introduced in many ways, some of which are the following:

Additive: $Y = F + EPS(1)$

Proportional: $Y = F * (1 + EPS(1))$

Exponential: $Y = \log(F) + EPS(1)$

For exponential, transform both sides by using log of the data in your data file and then code residual error model in control stream as

```
Y=LOG(F) +EPS(1)
```

The LOG() function in FORTRAN is natural logarithm

Additive and proportional residual error model:

```
Y = F*(1+EPS(1))+EPS(2)
```

If the problem is analytical and fairly simple, the user may completely define the model’s prediction function F and ETA and EPS relationships with the \$PRED block without using \$PK and \$ERROR blocks. In such cases, no ADVAN is used and \$SUBROUTINES is also not specified.

\$THETA block (chapter 9, guide V¹)—initial estimates and bounds of structural parameters

syntax:

<i>initial estimate</i>	e.g.	25.7
<i>(lower bound, initial estimate)</i>	e.g.	(0, 25.7)
<i>(lower bound, initial estimate, upper bound)</i>	e.g.	(0, 25.7, 120)
<i>(lower bound, upper bound)</i>	e.g.	(0, 120)
<i>(initial estimate FIXED)</i>	e.g.	(25.7 FIXED)

```

$THETA           ;comments are good programming practice
5.7              ;[KA]
(0, 6.2)         ;[CL]
(0,50.0, 120)   ;[VSS]
(0.0,,120.0)    ;[V]
(0.25 FIXED)    ;[ALAG1]

```

\$OMEGA block (chapter 9, guide V¹)—initial estimates of the variance of intersubject errors. The initial values of variances and covariances for the η (ETA) variables are expressed in a covariance matrix, Ω (OMEGA). The diagonal represents the variance and the lower triangle represents the covariance. Many analyses begin by assuming zeros for off-diagonal elements (this is the default setting). For a diagonal OMEGA matrix with two elements (ETA(1) and ETA(2)), with no correlation between ETAs,

```

$OMEGA
0.04 ;[P]
0.2 ;[A]

```

When using PDx-Pop with NONMEM, it is good to have an indicator to specify the type of variance model to assist interpretation for summary output, such as:

```

";[P]" for a proportional or exponential error model
";[A]" for an additive model
";[F]" for an off-diagonal covariance
";[N]" for not defined (zero)

```

For a matrix block, specify the lower triangular form:

```

$OMEGA BLOCK(3)
0.1 ; [p]
0.02 0.4 ; [p]
-0.01 0.02 0.2; [p]

```

\$SIGMA block (chapter 9, guide V¹)—initial estimates of the variance of residual errors variance estimates. The initial values of variances and covariances for the ϵ (EPS) residual deviates are expressed in a covariance matrix, Σ (SIGMA). As with **\$OMEGA**, the **\$SIGMA** may be structured as a block for modeling covariances between data types within a subject:

```

$SIGMA BLOCK(2)
0.1 ; [P]
0.02 0.4 ; [P]

```

Or more usually, independent SIGMA elements are modeled:

```

$SIGMA
0.05 ; [P]
1.0 ; [A]

```

(This example shows coding the initial estimates for the proportional and additive components of a combination residual error model.)

additive: $\text{var} = (\text{sd})^2$ e.g. if $\text{sd} = 0.1 \text{ nm/mL}$, $\text{var} = (0.1)^2 = 0.01$
 proportional or exponential (approximately): $\text{var} = (\text{cv}\%/100)^2$
 e.g., if $\text{cv}\% = 20$, $\text{var} = (20/100)^2 = 0.04$

\$ESTIMATION (chapters 9, 12, guide V¹)—sets methods and conditions for the estimation, may include METHOD, INTERACTION, PRINT, NOABORT, MAXEVAL, POSTHOC,

and other options. Here we only describe the options used for this example.

Always use INTER(ACTION) option. Without the INTERACTION option, the prediction F used in the multiplier expression to EPS() will be valuated at $\text{ETA} = 0$, which is contrary to how one should model data.

METHOD = 1 (or COND) indicates the first-order conditional estimation method is to be used. Because INTERACTION is also selected, the first-order conditional estimation with interaction is being performed in this example, which has become the standard classical method for NONMEM. Discussion of the various estimation methods available in NONMEM is provided in the advanced tutorial.

MAXEVAL=9999 indicates that up to 9,999 objective function evaluations are to be performed to obtain the best fit population parameters.

PRINT=5 indicates that every fifth iteration, the progress of the estimation should be reported.

Note that a FORMAT may be specified on the \$EST record for all auxiliary files that are automatically produced by NONMEM (see **Supplemental Materials S3-Part A, Additional Files Automatically Produced by NONMEM**) to control the delimiter and the precision of values outputted. For example, 1PE13.6 means comma delimited, E-field notation, one digit to left of decimal, and six digits to right of decimal, and field should not be > 13.

\$COVARIANCE (chapters 9, 12, guide V¹)—requests that standard errors of parameters and a correlation matrix should be generated.

\$TABLE (chapters 9, 10, guide V¹)—generates a table that includes:

Default items:

DV (dependent variable)

PRED (population prediction of dependent variable)

RES (residual) = DV – PRED

WRES (weighted residual) (CWRES is now preferred)

The user must specify: ID, TIME, individual predictions, any generated parameter, or data item in the data set that the user requests (ID and TIME are not default table items). This may include the list of all input variables and the labels of all estimated or calculated parameters

Include ONEHEADER, NOPRINT, and FILE=table.tab options. ONEHEADER means not to repeat the header (title and column labels) every 900 lines. NOPRINT means not to print the table output in the main report file but, rather, in the file specified with FILE=option. The *filename* may be any valid file name (example *filename* = 100.tab).

Note a FORMAT may be specified for each table to control the delimiter and the precision of values outputted. For example s1PE12.5 means space delimited, E-field notation, 1 digit to left of decimal, and 5 digits to right of decimal, and field should not be >12.

Running the program.

```
nmfe74 402.ct1 402.res -prdefault
```

Evaluating the output. View the NONMEM report file (402.res, full output file in **Supplementary Materials S1**) using your favorite text editor. Scroll down and check that

Table 2 Termination section of estimation in NONMEM report file

```

#TERM:
OMINIMIZATION SUCCESSFUL
NO. OF FUNCTION EVALUATIONS USED:      282
NO. OF SIG. DIGITS IN FINAL EST.:    3.5

ETABAR IS THE ARITHMETIC MEAN OF THE ETA-ESTIMATES,
AND THE P-VALUE IS GIVEN FOR THE NULL HYPOTHESIS THAT THE TRUE MEAN IS 0.

ETABAR:      -2.3376E-03  2.3646E-03  -2.6896E-03  -1.7738E-04
SE:          1.9718E-02  2.3009E-02  2.3703E-02  2.4590E-02
N:           30          30          30          30

P VAL.:      9.0563E-01  9.1815E-01  9.0966E-01  9.9424E-01

ETASHRINKSD(%) 9.9201E+00  9.3005E+00  2.2709E+01  1.0030E+01
ETASHRINKVR(%) 1.8856E+01  1.7736E+01  4.0260E+01  1.9054E+01
EBVSHRINKSD(%) 1.1325E+01  1.0912E+01  2.4240E+01  1.1491E+01
EBVSHRINKVR(%) 2.1367E+01  2.0633E+01  4.2604E+01  2.1662E+01
EPSSHRINKSD(%) 1.2856E+01
EPSSHRINKVR(%) 2.4058E+01

TOTAL DATA POINTS NORMALLY DISTRIBUTED (N):      360
N*LOG(2PI) CONSTANT TO OBJECTIVE FUNCTION:      661.635743907364
OBJECTIVE FUNCTION VALUE WITHOUT CONSTANT:      196.008341625908
OBJECTIVE FUNCTION VALUE WITH CONSTANT:      857.644085533273
REPORTED OBJECTIVE FUNCTION DOES NOT CONTAIN CONSTANT

TOTAL EFFECTIVE ETAS (NIND*NETA):      120

#TERE:
Elapsed estimation time in seconds:      1.45
Elapsed covariance time in seconds:      0.93
Elapsed postprocess time in seconds:      0.55

```

the run minimized successfully and no error messages exist (see portion of 402.res in **Table 2**). Error messages for estimation will appear between #TERM and #TERE. Additional error messages for covariance and table output assessment will appear after #TERE and before “postprocess time.”

Observe the minimum objective function (**Table 3**). Final THETA parameters, OMEGA parameters, and SIGMA parameters are also listed in the report file, along with the complete correlation form of OMEGA and SIGMA. The complete correlation form has standard deviations (SQRT(OMEGA)) on the diagonal positions and correlation on the off-diagonal positions.

Select a graphical software that will display the NONMEM table outputs. Programs such as Xpose,⁹ Pirana,³ PDx-Pop,² KIWI,⁵ and PLT Tools⁶ will present NONMEM graphical

results, typically using Excel or R. **Figures 2–4** were generated using R via PDx-Pop. In **Figure 2**, DV scatter pattern vs. PRED should approximately be balanced around the unity line unless the model is highly nonlinear. In **Figure 3**, the PRED values should be within the scatter regions of the DV data at all times. In **Figure 4**, the CWRES scatter should be symmetric about the 0 line, and most (95%) of the values should be between –2 and 2. This would suggest that the CWRES values are normally distributed, which occurs when the between-subject and within-subject statistical model, as well as the structural (PK/PD) model, properly represent the observed data.

\$COVARIANCE STEP OUTPUT

The \$COV step estimates the variance–covariance of the population parameters (THETA, OMEGA, SIGMA),

```

$TABLE ID TIME DV IPRE=CIPRED AMT CWRES IRES=CIRES IWRE=CIWRES
      NPd NPDE ESAMPLE=1000

```


Table 3 Objective function and final estimates in NONMEM report file

```

*****
FIRST ORDER CONDITIONAL ESTIMATION WITH INTERACTION
#OBJT:*****
MINIMUM VALUE OF OBJECTIVE FUNCTION
#OBJV:***** 196.008 *****
*****
FIRST ORDER CONDITIONAL ESTIMATION WITH INTERACTION
FINAL PARAMETER ESTIMATE
*****
*****
THETA - VECTOR OF FIXED EFFECTS PARAMETERS *****
TH 1 TH 2 TH 3 TH 4
9.76E+00 3.88E+00 3.08E+01 8.77E+00
OMEGA - COV MATRIX FOR RANDOM EFFECTS - ETAS *****
ETA1 ETA2 ETA3 ETA4
ETA1
+ 1.49E-02
ETA2
+ 0.00E+00 2.00E-02
ETA3
+ 0.00E+00 0.00E+00 2.92E-02
ETA4
+ 0.00E+00 0.00E+00 0.00E+00 2.32E-02
SIGMA - COV MATRIX FOR RANDOM EFFECTS - EPSILONS ****
EPS1
EPS1
+ 8.84E-03
OMEGA - CORR MATRIX FOR RANDOM EFFECTS - ETAS *****
ETA1 ETA2 ETA3 ETA4
ETA1
+ 1.22E-01
ETA2
+ 0.00E+00 1.41E-01
ETA3
+ 0.00E+00 0.00E+00 1.71E-01
ETA4
+ 0.00E+00 0.00E+00 0.00E+00 1.52E-01
SIGMA - CORR MATRIX FOR RANDOM EFFECTS - EPSILONS ***
EPS1
EPS1
9.40E-02

```




Figure 2 Run 402—DV (observations) vs. predicted concentrations with line of unity. The output from the user-specified file 402.tab is plotted by PDx-Pop using R, with filled circles representing the observed data against the population predictions PRED.




Figure 3 Run 402—DV and predicted concentrations vs. time. The output from the user-specified file 402.tab is plotted by PDx-Pop using R, with black filled circles representing the observed data and red filled circles representing the population prediction values PRED plotted against time.




Figure 4 Run 402—conditional weighted residuals (CWRES) vs. PRED. The output from the user-specified file 402.tab is plotted by PDx-Pop using R, with filled circles representing the CWRES values plotted against population prediction values PRED.

the square root of which serve as standard errors that are given in the report file along with the final estimates.

By dividing the standard error of the estimate by the final parameter estimate and multiplying by 100, a measure of the precision of the NONMEM parameter estimate, the relative standard error (RSE) is obtained:

$$RSE = 100 * (\text{Standard Error of Estimate}) / (\text{Estimate})$$

Symmetric 95% confidence intervals can also be calculated as the final estimate $\pm 1.96 \times$ standard error of the estimate.

Table 4 lists the standard error portion of the output report file for run 402. Notice that the standard errors to THETAs, OMEGAs, SIGMAs are provided and the complete correlation information (standard deviation on diagonal positions, correlation on off-diagonal positions) of the OMEGAs and SIGMAs are provided as well.

TABLE OUTPUTS (CHAPTERS 9, 10, GUIDE V¹)

As of NONMEM 7.3, individual predicted concentrations (CIPRED), residuals (CIRES), and weighted residuals (CIWRES) for the individual patients are produced automatically. Also, the population residual diagnostics CWRES, NPDE, and NPDE (decorrelated NPDE), may be outputted as well:

The ESAMPLE option specifies the number of random samples to generate the Monte Carlo based NPDE (correlated normalized prediction distribution error) and

Table 4 Standard errors of estimates in NONMEM report file

```

*****
*****
***** FIRST ORDER CONDITIONAL ESTIMATION WITH INTERACTION *****
***** STANDARD ERROR OF ESTIMATE *****
*****
THETA - VECTOR OF FIXED EFFECTS PARAMETERS *****
      TH 1      TH 2      TH 3      TH 4
      2.43E-01  1.07E-01  1.28E+00  2.73E-01

OMEGA - COV MATRIX FOR RANDOM EFFECTS - ETAS *****
      ETA1      ETA2      ETA3      ETA4
ETA1
+      3.42E-03
ETA2
+      .....  5.25E-03
ETA3
+      .....  .....  1.12E-02
ETA4
+      .....  .....  .....  5.65E-03

SIGMA - COV MATRIX FOR RANDOM EFFECTS - EPSILONS ****
      EPS1
      6.59E-04
      1

OMEGA - CORR MATRIX FOR RANDOM EFFECTS - ETAS *****
      ETA1      ETA2      ETA3      ETA4
ETA1
+      1.40E-02
ETA2
+      .....  1.86E-02
ETA3
+      .....  .....  3.27E-02
ETA4
+      .....  .....  .....  1.86E-02

SIGMA - CORR MATRIX FOR RANDOM EFFECTS - EPSILONS ***
      EPS1
      3.50E-03

```

```
$PROB Phase IIa Study, One Compartment Model 504.ct1
; Place column names of data file here:
$INPUT C ID TIME DV AMT RATE WT AGE SEX
$DATA 501.csv IGNORE=C ; Ignore records beginning with letter C
; Select One compartment model ADVAN1, Parameterization TRANS2 (CL, V)
$SUBROUTINE ADVAN1 TRANS2
; Section to define PK parameters, relationship to fixed effects THETA
; and inter-subject random effects ETA.
$PK
; Define typical values
TVCL=THETA (1) * (WT/70) **THETA (3) * (AGE/50) **THETA (5) *THETA (7) **SEX
TVV=THETA (2) * (WT/70) **THETA (4) * (AGE/50) **THETA (6) *THETA (8) **SEX
CL=TVCL*EXP (ETA (1))
V=TVV*EXP (ETA (2))
S1=V
$THETA ; Enter initial starting values for THETAS
(0,4) ; [CL]
(0,30) ; [V]
0.8 ; [CL~WT]
0.8 ; [V~WT]
-0.1 ; [CL~AGE]
0.1 ; [V~AGE]
0.7 ; [CL~SEX]
0.7 ; [V~SEX]

; Section to relate predicted function F and residual error
; relationship to data DV. EPS are random error coefficients
$ERROR
Y=F*(1+EPS(1))
$OMEGA BLOCK(2) ; Initial OMEGA values in lower triangular format
0.1 ; [P]
0.001 0.1 ; [P]
$SIGMA ; Initial SIGMA
0.04 ; [P]

;FOCEI is selected
$EST METHOD=COND INTERACTION MAXEVAL=9999 PRINT=5 NOABORT
; Evaluate variance-covariance of estimates
$COV UNCONDITIONAL MATRIX=R PRINT=E
; Print out individual predicted results and diagnostics
; to file 504.tab
; Various parameters and built in diagnostics may be printed.
; DV=DEPENDENT VARIABLE
; CIPRED=individual predicted function, f(eta_hat), at mode of
; conditional density
; CIRES=DV-F(ETA_HAT)
; CIWRES=conditional individual residual
; (DV-F(ETA_HAT))/SQRT(SIGMA(1,1)*F(ETA_HAT))
; PRED=Population Predicted value F(ETA=0)
; CWRES=Population weighted Residual
; Note numerical Format may be specified for table outputs
$TABLE ID TIME DV CIPRED CIRES CIWRES PRED RES CWRES CL V ETA1 ETA2
NOPRINT NOAPPEND ONEHEADER FORMAT=,1PE13.6 FILE=504.tab
```

NPDE (normalized prediction distribution error) diagnostics. Additional individual parameters, as defined in the \$PK block, may also be requested (complete table 402.tab is in **Supplemental Materials S1**):

CL V1 Q V2 ETA1 ETA2 ETA3 ETA4

See **Supplemental Materials S3-Part A** for additional output files produced by NONMEM. Also, more detailed information for the various methods and options can be

obtained from the NONMEM user guides. A guide to the NONMEM user's guides is listed in **Supplemental Materials S3-Part B**.

STATISTICAL ASSESSMENT OF COVARIATES EFFECTS ADDED TO A PK MODEL

One of the goals of population analysis is to explain as much between-subject variability in the model parameters as possible. The two main ways to account for the between-subject variability are to add covariate effects to the model in an attempt to reduce the unexplained variability and to reparameterize the between-subject variability model.

In this example (504.ct1), 60 surgical patients in a phase IIa study received a 1 mg, 2 mg, or 4 mg dose of an anesthetic over a 5-hour constant-rate intravenous infusion. For each individual, plasma samples were obtained at four time points after the start of the infusion (1, 5, 12, and 24 hours postinitiation). The drug assay has a constant coefficient of variation as concentrations increase. This data set includes patient demographic data (weight, age, sex), which was used to determine the covariate relationships with the PK parameters. The data were fit to a one-compartment constant-rate infusion model with the parameters CL and V. The control stream file may be constructed for first-order conditional estimation with interaction using the ADVAN1 library module in NONMEM, as shown here:

Notice in the \$PK block that the covariates are included in the model as predictors of the "typical value" parameters TVCL (typical value clearance), and TVV typical value volume, (bolded text). Furthermore, the covariate is divided by a value that represents its average or median of the data set (such as median age = 50) or average for a typical population (70 kg for body weight). In this way, the THETAs representing the base clearance CL or volume V (THETA(1), THETA(2)) are readily interpreted as the typical clearance or volume for a subject of typical (average or median) body weight or age. This is followed by expressing the individual parameters CL and V as a function of the random effects ETA(), multiplicatively centered about the typical values, because the exponentials of the ETAs are used. Because the ETAs are normally distributed with variances of OMEGA, the individual parameters are log-normally distributed; that is, the logarithms of the individual parameters are normally distributed. This is a typical manner in which population analysis is performed, as it provides a straightforward means of incorporating the fixed effects model with covariates followed by the expression of the individual parameters, and the log-normal format assures that the individual parameters are always positive. If individual parameters are modeled as a log of normal distribution, it is often convenient to use the power model (COV**THETA(x)) to incorporate continuous covariate effects to the typical value, which when log transformed, presents the covariate coefficients THETA(x) as a linear effect on the log of the typical values of the parameters.

Notice that the effect of SEX, a categorical covariate, is modeled as THETA(x)**SEX. Thus when SEX=0,

$$\text{THETA}(1) \dots \text{THETA}(x) ** \text{SEX} = \text{THETA}(1) \dots$$

and when SEX=1,

$$\text{THETA}(1) \dots \text{THETA}(x) ** \text{SEX} = \text{THETA}(1) \dots \text{THETA}(x).$$

That is, SEX=0 has the typical value, and SEX=1 has the typical value modified by THETA(x).

```
$THETA ; Enter initial starting values for THETAS
(0,4) ; [CL]
(0,30) ; [V]
(0.75 FIXED) ; [CL~WT]
(1.0 FIXED) ; [V~WT]
-0.1 ; [CL~AGE]
(0.0 FIXED) ; [V~AGE]
(1.0 FIXED) ; [CL~SEX]
(1.0 FIXED) ; [V~SEX]
```

Upon review of the 504.res file from this model (complete report file in **Supplementary Materials S2**), it is noted that the model could be simplified by recognizing that the parameters estimating the effects of SEX on CL and V are both estimated to be near 1, the effect of AGE on V is near 0, and the effect of weight WT on CL is near 0.75 (an allometric weight coefficient that is consistent with literature for allometric relationships between CL and WT for many small molecules). The final parameter estimates and corresponding 95% confidence intervals for each of these parameters is shown below:

Covariate	Estimate	95% confidence intervals	
CL~ SEX	0.91	0.80–1.01	Not clinically significant
V~ SEX	0.95	0.81–1.08	No change in V
V~ AGE	0.06	–0.21 to 0.31	No clinically significant trend
CL~ WT	0.66	0.36–0.95	Consistent with allometric coefficient (0.75)
V~ WT	1.32	0.95–1.69	Consistent with allometric coefficient (1.0)

Rather than modifying the typical values equations in \$PK, the PK model can be more conveniently and reversibly simplified by fixing the appropriate THETAs (504f.ct1, **Supplementary Materials S2**):

resulting in an objective function of 1065.362 when assessed by first-order conditional estimation with interaction, seven units larger than the full model, not statistically different when assessing the objective function difference as a chi-square statistic with five degrees of freedom.

In the advanced tutorial on NONMEM, this example is embellished to incorporate Monte Carlo EM and Bayesian methods of analysis.

Supporting Information. Supplementary information accompanies this paper on the *CPT: Pharmacometrics & Systems Pharmacology* website (www.psp-journal.com).

Supplementary Materials S1. 402 example (file NONMEM_Tutorial_I_S1.zip).

Supplementary Materials S2. 504 example (file NONMEM_Tutorial_I_S2.zip).

Supplementary Materials S3. NONMEM files and user guide (file S3_Parts_A_and_B).

Acknowledgments. Portions of this manuscript are derived from the ICON PDX-Pop and NONMEM workshop. William J. Bachman, Ph.D., Ekaterina Gibiansky, Ph.D., Brian Sadler, Ph.D., Indrajit Das, Ph.D., and Robert J. Bauer, Ph.D., Editors, ICON plc.

Funding. ICON Clinical Research LLC. licenses the NONMEM and PDX-Pop software.

Conflict of Interest. The author is a paid employee of ICON Clinical Research LLC. and is the present developer of NONMEM.

1. Beal, S.L., Sheiner, L.B., Boeckmann, A.J. & Bauer, R.J. (Eds). NONMEM 7.4 users guides <<https://nonmem.iconplc.com/nonmem743/guides>> (1989–2018).
2. Das, I. & Bachman, W. PDX-Pop user manual version 5.2.2 <<https://nonmem.iconplc.com/pdxpop/PDX-Pop5.2.2>> (2018).
3. Pirana installation guide and manual, version 2.10.0 <<http://www.pirana-software.com/pub/docs/Manual.pdf>> (2015).
4. Mats, K., Rikard, N., Svetlana, F., Sebastian, U. & Gunnar, Y. Pearl speaks NONMEM version 4.8.1 <<https://uupharmacometrics.github.io/PsN/>> (2018).

5. Cognigen, Simulations Plus. KIWI 3.0 <<https://www.simulations-plus.com/software/kiwi/>> (2013).
6. Fisher, D. & Shafer, S. PLT tools <<http://www.pltsoft.com>> (2017).
7. Holford, N. Wings for NONMEM <<http://wfn.sourceforge.net/>> (2017).
8. Owen, J.S. & Fiedler-Kelly, J. Introduction to Population Pharmacokinetic/Pharmacodynamic Analysis with Nonlinear Mixed Effects Models *1st ed.* (John Wiley & Sons, Hoboken, NJ, 2014).
9. Hooker, A.C., Karlsson, M.O. & Jonsson, E.N. 2009–2014 Xpose version 4.5.3 <<http://xpose.sourceforge.net/>> (2014).

© 2019 ICON Clinical Research LLC. CPT: *Pharmacometrics & Systems Pharmacology* published by Wiley Periodicals, Inc. on behalf of the American Society for Clinical Pharmacology & Therapeutics. This is an open access article under the terms of the Creative Commons Attribution-NonCommercial-NoDerivs License, which permits use and distribution in any medium, provided the original work is properly cited, the use is non-commercial and no modifications or adaptations are made.